

# NAG Library Routine Document

## G08BAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G08BAF performs Mood's and David's tests for dispersion differences between two independent samples of possibly unequal size.

### 2 Specification

```
SUBROUTINE G08BAF (X, N, N1, R, ITEST, W, V, PW, PV, IFAIL)
```

```
INTEGER          N, N1, ITEST, IFAIL
REAL (KIND=nag_wp) X(N), R(N), W, V, PW, PV
```

### 3 Description

Mood's and David's tests investigate the difference between the dispersions of two independent samples of sizes  $n_1$  and  $n_2$ , denoted by

$$x_1, x_2, \dots, x_{n_1}$$

and

$$x_{n_1+1}, x_{n_1+2}, \dots, x_n, \quad n = n_1 + n_2.$$

The hypothesis under test,  $H_0$ , often called the null hypothesis, is that the dispersion difference is zero, and this is to be tested against a one- or two-sided alternative hypothesis  $H_1$  (see below).

Both tests are based on the rankings of the sample members within the pooled sample formed by combining both samples. If there is some difference in dispersion, more of the extreme ranks will tend to be found in one sample than in the other.

Let the rank of  $x_i$  be denoted by  $r_i$ , for  $i = 1, 2, \dots, n$ .

(a) Mood's test.

The test statistic  $W = \sum_{i=1}^{n_1} \left( r_i - \frac{n+1}{2} \right)^2$  is found.

$W$  is the sum of squared deviations from the average rank in the pooled sample. For large  $n$ ,  $W$  approaches normality, and so an approximation,  $p_w$ , to the probability of observing  $W$  not greater than the computed value, may be found.

G08BAF returns  $W$  and  $p_w$  if Mood's test is selected.

(b) David's test.

The disadvantage of Mood's test is that it assumes that the means of the two samples are equal. If this assumption is unjustified a high value of  $W$  could merely reflect the difference in means. David's test reduces this effect by using the variance of the ranks of the first sample about their mean rank, rather than the overall mean rank.

The test statistic for David's test is

$$V = \frac{1}{n_1 - 1} \sum_{i=1}^{n_1} (r_i - \bar{r})^2$$

where

$$\bar{r} = \frac{\sum_{i=1}^{n_1} r_i}{n_1}.$$

For large  $n$ ,  $V$  approaches normality, enabling an approximate probability  $p_v$  to be computed, similarly to  $p_w$ .

G08BAF returns  $V$  and  $p_v$  if David's test is selected.

Suppose that a significance test of a chosen size  $\alpha$  is to be performed (i.e.,  $\alpha$  is the probability of rejecting  $H_0$  when  $H_0$  is true; typically  $\alpha$  is a small quantity such as 0.05 or 0.01).

The returned value  $p$  ( $= p_v$  or  $p_w$ ) can be used to perform a significance test, against various alternative hypotheses  $H_1$ , as follows.

- (i)  $H_1$ : dispersions are unequal.  $H_0$  is rejected if  $2 \times \min(p, 1 - p) < \alpha$ .
- (ii)  $H_1$ : dispersion of sample 1  $>$  dispersion of sample 2.  $H_0$  is rejected if  $1 - p < \alpha$ .
- (iii)  $H_1$ : dispersion of sample 2  $>$  dispersion of sample 1.  $H_0$  is rejected if  $p < \alpha$ .

## 4 References

Cooper B E (1975) *Statistics for Experimentalists* Pergamon Press

## 5 Parameters

- |    |   |               |
|----|---|---------------|
| 1: | X(N) – REAL (KIND=nag_wp) array   | <i>Input</i>  |
|    | <i>On entry:</i> the first $n_1$ elements of X must be set to the data values in the first sample, and the next $n_2$ ( $= N - n_1$ ) elements to the data values in the second sample. |               |
| 2: | N – INTEGER   | <i>Input</i>  |
|    | <i>On entry:</i> the total of the two sample sizes, $n$ ( $= n_1 + n_2$ ).  |               |
|    | <i>Constraint:</i> $N > 2$ .  |               |
| 3: | N1 – INTEGER  | <i>Input</i>  |
|    | <i>On entry:</i> the size of the first sample, $n_1$ .  |               |
|    | <i>Constraint:</i> $1 < N1 < N$ .   |               |
| 4: | R(N) – REAL (KIND=nag_wp) array   | <i>Output</i> |
|    | <i>On exit:</i> the ranks $r_i$ , assigned to the data values $x_i$ , for $i = 1, 2, \dots, n$ .  |               |
| 5: | ITEST – INTEGER   | <i>Input</i>  |
|    | <i>On entry:</i> the test(s) to be carried out.   |               |
|    | ITEST = 0<br>Both Mood's and David's tests.   |               |
|    | ITEST = 1<br>David's test only.   |               |
|    | ITEST = 2<br>Mood's test only.  |               |
|    | <i>Constraint:</i> ITEST = 0, 1 or 2.   |               |
| 6: | W – REAL (KIND=nag_wp)  | <i>Output</i> |
|    | <i>On exit:</i> Mood's test statistic, $W$ , if requested.  |               |

- 7: V – REAL (KIND=nag\_wp) Output  
*On exit:* David's test statistic,  $V$ , if requested.
- 8: PW – REAL (KIND=nag\_wp) Output  
*On exit:* the lower tail probability,  $p_w$ , corresponding to the value of  $W$ , if Mood's test was requested.
- 9: PV – REAL (KIND=nag\_wp) Output  
*On exit:* the lower tail probability,  $p_v$ , corresponding to the value of  $V$ , if David's test was requested.
- 10: IFAIL – INTEGER Input/Output  
*On entry:* IFAIL must be set to 0,  $-1$  or  $1$ . If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
- For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or  $1$  is recommended. If the output of error messages is undesirable, then the value  $1$  is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is  $0$ . **When the value  $-1$  or  $1$  is used it is essential to test the value of IFAIL on exit.**
- On exit:* IFAIL =  $0$  unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL =  $0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N \leq 2$ .

IFAIL = 2

On entry,  $N1 \leq 1$ ,  
 or  $N1 \geq N$ .

IFAIL = 3

On entry, ITEST <  $0$ ,  
 or ITEST >  $2$ .

## 7 Accuracy

All computations are believed to be stable. The statistics  $V$  and  $W$  should be accurate enough for all practical uses.

## 8 Further Comments

The time taken by G08BAF is small, and increases with  $n$ .

## 9 Example

This example is taken from page 280 of Cooper (1975). The data consists of two samples of six observations each. Both Mood's and David's test statistics and significances are computed. Note that Mood's statistic is inflated owing to the difference in location of the two samples, the median ranks differing by a factor of two.

## 9.1 Program Text

```

Program g08bafe

!      G08BAF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
Use nag_library, Only: g08baf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: pv, pw, v, w
Integer                    :: ifail, itest, n, n1
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: r(:), x(:)
!      .. Executable Statements ..
Write (nout,*) 'G08BAF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in problem size and type of test
Read (nin,*) n, n1, itest

Allocate (x(n),r(n))

!      Read in data
Read (nin,*) x(1:n)

!      Display title
Write (nout,*) 'Mood''s test and David''s test'
Write (nout,*)

!      Display input data
Write (nout,*) 'Data values'
Write (nout,*)
Write (nout,*) 'Group 1  '
Write (nout,99999) x(1:n1)
Write (nout,*)
Write (nout,*) 'Group 2  '
Write (nout,99999) x((n1+1):n)

!      Perform test
ifail = 0
Call g08baf(x,n,n1,r,itest,w,v,pw,pv,ifail)

!      Display results
Write (nout,*)
Write (nout,99998) '      Mood''s measure = ', w, '      Significance = ', &
  pw
Write (nout,99998) '      David''s measure = ', v, '      Significance = ', &
  pv

99999 Format (1X,8F4.0)
99998 Format (1X,A,F8.3,A,F8.4)
End Program g08bafe

```

## 9.2 Program Data

```

G08BAF Example Program Data
12 6 0          :: N,N1,ITEST
 6.0 9.0 12.0 4.0 10.0 11.0
 8.0 1.0 3.0 7.0 2.0 5.0 :: End of X

```

### 9.3 Program Results

G08BAF Example Program Results

Mood's test and David's test

Data values

Group 1

6. 9. 12. 4. 10. 11.

Group 2

8. 1. 3. 7. 2. 5.

Mood's measure =	75.500	Significance =	0.5830
David's measure =	9.467	Significance =	0.1986

---